creation_sourcecode: // SPDX-License-Identifier: UNLICENSED
  pragma solidity 0.8.26;
  
  /**
   * @dev Interface of the ERC20 standard as defined in the EIP.
   */
  interface IERC20 {
      /**
       * @dev Emitted when `value` tokens are moved from one account (`from`) to
       * another (`to`).
       *
       * Note that `value` may be zero.
       */
      event Transfer(address indexed from, address indexed to, uint256 value);
  
      /**
       * @dev Emitted when the allowance of a `spender` for an `owner` is set by
       * a call to {approve}. `value` is the new allowance.
       */
      event Approval(address indexed owner, address indexed spender, uint256 value);
  
      /**
       * @dev Returns the value of tokens in existence.
       */
      function totalSupply() external view returns (uint256);
  
      /**
       * @dev Returns the value of tokens owned by `account`.
       */
      function balanceOf(address account) external view returns (uint256);
  
      /**
       * @dev Moves a `value` amount of tokens from the caller's account to `to`.
       *
       * Returns a boolean value indicating whether the operation succeeded.
       *
       * Emits a {Transfer} event.
       */
      function transfer(address to, uint256 value) external returns (bool);
  
      /**
       * @dev Returns the remaining number of tokens that `spender` will be
       * allowed to spend on behalf of `owner` through {transferFrom}. This is
       * zero by default.
       *
       * This value changes when {approve} or {transferFrom} are called.
       */
      function allowance(address owner, address spender) external view returns (uint256);
  
      /**
       * @dev Sets a `value` amount of tokens as the allowance of `spender` over the
       * caller's tokens.
       *
       * Returns a boolean value indicating whether the operation succeeded.
       *
       * IMPORTANT: Beware that changing an allowance with this method brings the risk
       * that someone may use both the old and the new allowance by unfortunate
       * transaction ordering. One possible solution to mitigate this race
       * condition is to first reduce the spender's allowance to 0 and set the
       * desired value afterwards:
       * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
       *
       * Emits an {Approval} event.
       */
      function approve(address spender, uint256 value) external returns (bool);
  
      /**
       * @dev Moves a `value` amount of tokens from `from` to `to` using the
       * allowance mechanism. `value` is then deducted from the caller's
       * allowance.
       *
       * Returns a boolean value indicating whether the operation succeeded.
       *
       * Emits a {Transfer} event.
       */
      function transferFrom(address from, address to, uint256 value) external returns (bool);
  }
  
  /**
   * @dev Interface for the optional metadata functions from the ERC20 standard.
   */
  interface IERC20Metadata is IERC20 {
      /**
       * @dev Returns the name of the token.
       */
      function name() external view returns (string memory);
  
      /**
       * @dev Returns the symbol of the token.
       */
      function symbol() external view returns (string memory);
  
      /**
       * @dev Returns the decimals places of the token.
       */
      function decimals() external view returns (uint8);
  }
  
  /**
   * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
   * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
   *
   * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
   * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
   * need to send a transaction, and thus is not required to hold Ether at all.
   *
   * ==== Security Considerations
   *
   * There are two important considerations concerning the use of `permit`. The first is that a valid permit signature
   * expresses an allowance, and it should not be assumed to convey additional meaning. In particular, it should not be
   * considered as an intention to spend the allowance in any specific way. The second is that because permits have
   * built-in replay protection and can be submitted by anyone, they can be frontrun. A protocol that uses permits should
   * take this into consideration and allow a `permit` call to fail. Combining these two aspects, a pattern that may be
   * generally recommended is:
   *
   * ```solidity
   * function doThingWithPermit(..., uint256 value, uint256 deadline, uint8 v, bytes32 r, bytes32 s) public {
   *     try token.permit(msg.sender, address(this), value, deadline, v, r, s) {} catch {}
   *     doThing(..., value);
   * }
   *
   * function doThing(..., uint256 value) public {
   *     token.safeTransferFrom(msg.sender, address(this), value);
   *     ...
   * }
   * ```
   *
   * Observe that: 1) `msg.sender` is used as the owner, leaving no ambiguity as to the signer intent, and 2) the use of
   * `try/catch` allows the permit to fail and makes the code tolerant to frontrunning. (See also
   * {SafeERC20-safeTransferFrom}).
   *
   * Additionally, note that smart contract wallets (such as Argent or Safe) are not able to produce permit signatures, so
   * contracts should have entry points that don't rely on permit.
   */
  interface IERC20Permit {
      /**
       * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
       * given ``owner``'s signed approval.
       *
       * IMPORTANT: The same issues {IERC20-approve} has related to transaction
       * ordering also apply here.
       *
       * Emits an {Approval} event.
       *
       * Requirements:
       *
       * - `spender` cannot be the zero address.
       * - `deadline` must be a timestamp in the future.
       * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
       * over the EIP712-formatted function arguments.
       * - the signature must use ``owner``'s current nonce (see {nonces}).
       *
       * For more information on the signature format, see the
       * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
       * section].
       *
       * CAUTION: See Security Considerations above.
       */
      function permit(
          address owner,
          address spender,
          uint256 value,
          uint256 deadline,
          uint8 v,
          bytes32 r,
          bytes32 s
      ) external;
  
      /**
       * @dev Returns the current nonce for `owner`. This value must be
       * included whenever a signature is generated for {permit}.
       *
       * Every successful call to {permit} increases ``owner``'s nonce by one. This
       * prevents a signature from being used multiple times.
       */
      function nonces(address owner) external view returns (uint256);
  
      /**
       * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
       */
      // solhint-disable-next-line func-name-mixedcase
      function DOMAIN_SEPARATOR() external view returns (bytes32);
  }
  
  /**
   * @dev Collection of functions related to the address type
   */
  library Address {
      /**
       * @dev The ETH balance of the account is not enough to perform the operation.
       */
      error AddressInsufficientBalance(address account);
  
      /**
       * @dev There's no code at `target` (it is not a contract).
       */
      error AddressEmptyCode(address target);
  
      /**
       * @dev A call to an address target failed. The target may have reverted.
       */
      error FailedInnerCall();
  
      /**
       * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
       * `recipient`, forwarding all available gas and reverting on errors.
       *
       * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
       * of certain opcodes, possibly making contracts go over the 2300 gas limit
       * imposed by `transfer`, making them unable to receive funds via
       * `transfer`. {sendValue} removes this limitation.
       *
       * https://consensys.net/diligence/blog/2019/09/stop-using-soliditys-transfer-now/[Learn more].
       *
       * IMPORTANT: because control is transferred to `recipient`, care must be
       * taken to not create reentrancy vulnerabilities. Consider using
       * {ReentrancyGuard} or the
       * https://solidity.readthedocs.io/en/v0.8.20/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
       */
      function sendValue(address payable recipient, uint256 amount) internal {
          if (address(this).balance < amount) {
              revert AddressInsufficientBalance(address(this));
          }
  
          (bool success, ) = recipient.call{value: amount}("");
          if (!success) {
              revert FailedInnerCall();
          }
      }
  
      /**
       * @dev Performs a Solidity function call using a low level `call`. A
       * plain `call` is an unsafe replacement for a function call: use this
       * function instead.
       *
       * If `target` reverts with a revert reason or custom error, it is bubbled
       * up by this function (like regular Solidity function calls). However, if
       * the call reverted with no returned reason, this function reverts with a
       * {FailedInnerCall} error.
       *
       * Returns the raw returned data. To convert to the expected return value,
       * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
       *
       * Requirements:
       *
       * - `target` must be a contract.
       * - calling `target` with `data` must not revert.
       */
      function functionCall(address target, bytes memory data) internal returns (bytes memory) {
          return functionCallWithValue(target, data, 0);
      }
  
      /**
       * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
       * but also transferring `value` wei to `target`.
       *
       * Requirements:
       *
       * - the calling contract must have an ETH balance of at least `value`.
       * - the called Solidity function must be `payable`.
       */
      function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
          if (address(this).balance < value) {
              revert AddressInsufficientBalance(address(this));
          }
          (bool success, bytes memory returndata) = target.call{value: value}(data);
          return verifyCallResultFromTarget(target, success, returndata);
      }
  
      /**
       * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
       * but performing a static call.
       */
      function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
          (bool success, bytes memory returndata) = target.staticcall(data);
          return verifyCallResultFromTarget(target, success, returndata);
      }
  
      /**
       * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
       * but performing a delegate call.
       */
      function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
          (bool success, bytes memory returndata) = target.delegatecall(data);
          return verifyCallResultFromTarget(target, success, returndata);
      }
  
      /**
       * @dev Tool to verify that a low level call to smart-contract was successful, and reverts if the target
       * was not a contract or bubbling up the revert reason (falling back to {FailedInnerCall}) in case of an
       * unsuccessful call.
       */
      function verifyCallResultFromTarget(
          address target,
          bool success,
          bytes memory returndata
      ) internal view returns (bytes memory) {
          if (!success) {
              _revert(returndata);
          } else {
              // only check if target is a contract if the call was successful and the return data is empty
              // otherwise we already know that it was a contract
              if (returndata.length == 0 && target.code.length == 0) {
                  revert AddressEmptyCode(target);
              }
              return returndata;
          }
      }
  
      /**
       * @dev Tool to verify that a low level call was successful, and reverts if it wasn't, either by bubbling the
       * revert reason or with a default {FailedInnerCall} error.
       */
      function verifyCallResult(bool success, bytes memory returndata) internal pure returns (bytes memory) {
          if (!success) {
              _revert(returndata);
          } else {
              return returndata;
          }
      }
  
      /**
       * @dev Reverts with returndata if present. Otherwise reverts with {FailedInnerCall}.
       */
      function _revert(bytes memory returndata) private pure {
          // Look for revert reason and bubble it up if present
          if (returndata.length > 0) {
              // The easiest way to bubble the revert reason is using memory via assembly
              /// @solidity memory-safe-assembly
              assembly {
                  let returndata_size := mload(returndata)
                  revert(add(32, returndata), returndata_size)
              }
          } else {
              revert FailedInnerCall();
          }
      }
  }
  
  /**
   * @title SafeERC20
   * @dev Wrappers around ERC20 operations that throw on failure (when the token
   * contract returns false). Tokens that return no value (and instead revert or
   * throw on failure) are also supported, non-reverting calls are assumed to be
   * successful.
   * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
   * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
   */
  library SafeERC20 {
      using Address for address;
  
      /**
       * @dev An operation with an ERC20 token failed.
       */
      error SafeERC20FailedOperation(address token);
  
      /**
       * @dev Indicates a failed `decreaseAllowance` request.
       */
      error SafeERC20FailedDecreaseAllowance(address spender, uint256 currentAllowance, uint256 requestedDecrease);
  
      /**
       * @dev Transfer `value` amount of `token` from the calling contract to `to`. If `token` returns no value,
       * non-reverting calls are assumed to be successful.
       */
      function safeTransfer(IERC20 token, address to, uint256 value) internal {
          _callOptionalReturn(token, abi.encodeCall(token.transfer, (to, value)));
      }
  
      /**
       * @dev Transfer `value` amount of `token` from `from` to `to`, spending the approval given by `from` to the
       * calling contract. If `token` returns no value, non-reverting calls are assumed to be successful.
       */
      function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
          _callOptionalReturn(token, abi.encodeCall(token.transferFrom, (from, to, value)));
      }
  
      /**
       * @dev Increase the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
       * non-reverting calls are assumed to be successful.
       */
      function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
          uint256 oldAllowance = token.allowance(address(this), spender);
          forceApprove(token, spender, oldAllowance + value);
      }
  
      /**
       * @dev Decrease the calling contract's allowance toward `spender` by `requestedDecrease`. If `token` returns no
       * value, non-reverting calls are assumed to be successful.
       */
      function safeDecreaseAllowance(IERC20 token, address spender, uint256 requestedDecrease) internal {
          unchecked {
              uint256 currentAllowance = token.allowance(address(this), spender);
              if (currentAllowance < requestedDecrease) {
                  revert SafeERC20FailedDecreaseAllowance(spender, currentAllowance, requestedDecrease);
              }
              forceApprove(token, spender, currentAllowance - requestedDecrease);
          }
      }
  
      /**
       * @dev Set the calling contract's allowance toward `spender` to `value`. If `token` returns no value,
       * non-reverting calls are assumed to be successful. Meant to be used with tokens that require the approval
       * to be set to zero before setting it to a non-zero value, such as USDT.
       */
      function forceApprove(IERC20 token, address spender, uint256 value) internal {
          bytes memory approvalCall = abi.encodeCall(token.approve, (spender, value));
  
          if (!_callOptionalReturnBool(token, approvalCall)) {
              _callOptionalReturn(token, abi.encodeCall(token.approve, (spender, 0)));
              _callOptionalReturn(token, approvalCall);
          }
      }
  
      /**
       * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
       * on the return value: the return value is optional (but if data is returned, it must not be false).
       * @param token The token targeted by the call.
       * @param data The call data (encoded using abi.encode or one of its variants).
       */
      function _callOptionalReturn(IERC20 token, bytes memory data) private {
          // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
          // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
          // the target address contains contract code and also asserts for success in the low-level call.
  
          bytes memory returndata = address(token).functionCall(data);
          if (returndata.length != 0 && !abi.decode(returndata, (bool))) {
              revert SafeERC20FailedOperation(address(token));
          }
      }
  
      /**
       * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
       * on the return value: the return value is optional (but if data is returned, it must not be false).
       * @param token The token targeted by the call.
       * @param data The call data (encoded using abi.encode or one of its variants).
       *
       * This is a variant of {_callOptionalReturn} that silents catches all reverts and returns a bool instead.
       */
      function _callOptionalReturnBool(IERC20 token, bytes memory data) private returns (bool) {
          // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
          // we're implementing it ourselves. We cannot use {Address-functionCall} here since this should return false
          // and not revert is the subcall reverts.
  
          (bool success, bytes memory returndata) = address(token).call(data);
          return success && (returndata.length == 0 || abi.decode(returndata, (bool))) && address(token).code.length > 0;
      }
  }
  
  /**
   * @dev Standard math utilities missing in the Solidity language.
   */
  library Math {
      /**
       * @dev Muldiv operation overflow.
       */
      error MathOverflowedMulDiv();
  
      enum Rounding {
          Floor, // Toward negative infinity
          Ceil, // Toward positive infinity
          Trunc, // Toward zero
          Expand // Away from zero
      }
  
      /**
       * @dev Returns the addition of two unsigned integers, with an overflow flag.
       */
      function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
          unchecked {
              uint256 c = a + b;
              if (c < a) return (false, 0);
              return (true, c);
          }
      }
  
      /**
       * @dev Returns the subtraction of two unsigned integers, with an overflow flag.
       */
      function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
          unchecked {
              if (b > a) return (false, 0);
              return (true, a - b);
          }
      }
  
      /**
       * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
       */
      function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
          unchecked {
              // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
              // benefit is lost if 'b' is also tested.
              // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
              if (a == 0) return (true, 0);
              uint256 c = a * b;
              if (c / a != b) return (false, 0);
              return (true, c);
          }
      }
  
      /**
       * @dev Returns the division of two unsigned integers, with a division by zero flag.
       */
      function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
          unchecked {
              if (b == 0) return (false, 0);
              return (true, a / b);
          }
      }
  
      /**
       * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
       */
      function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
          unchecked {
              if (b == 0) return (false, 0);
              return (true, a % b);
          }
      }
  
      /**
       * @dev Returns the largest of two numbers.
       */
      function max(uint256 a, uint256 b) internal pure returns (uint256) {
          return a > b ? a : b;
      }
  
      /**
       * @dev Returns the smallest of two numbers.
       */
      function min(uint256 a, uint256 b) internal pure returns (uint256) {
          return a < b ? a : b;
      }
  
      /**
       * @dev Returns the average of two numbers. The result is rounded towards
       * zero.
       */
      function average(uint256 a, uint256 b) internal pure returns (uint256) {
          // (a + b) / 2 can overflow.
          return (a & b) + (a ^ b) / 2;
      }
  
      /**
       * @dev Returns the ceiling of the division of two numbers.
       *
       * This differs from standard division with `/` in that it rounds towards infinity instead
       * of rounding towards zero.
       */
      function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
          if (b == 0) {
              // Guarantee the same behavior as in a regular Solidity division.
              return a / b;
          }
  
          // (a + b - 1) / b can overflow on addition, so we distribute.
          return a == 0 ? 0 : (a - 1) / b + 1;
      }
  
      /**
       * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or
       * denominator == 0.
       * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv) with further edits by
       * Uniswap Labs also under MIT license.
       */
      function mulDiv(uint256 x, uint256 y, uint256 denominator) internal pure returns (uint256 result) {
          unchecked {
              // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
              // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
              // variables such that product = prod1 * 2^256 + prod0.
              uint256 prod0 = x * y; // Least significant 256 bits of the product
              uint256 prod1; // Most significant 256 bits of the product
              assembly {
                  let mm := mulmod(x, y, not(0))
                  prod1 := sub(sub(mm, prod0), lt(mm, prod0))
              }
  
              // Handle non-overflow cases, 256 by 256 division.
              if (prod1 == 0) {
                  // Solidity will revert if denominator == 0, unlike the div opcode on its own.
                  // The surrounding unchecked block does not change this fact.
                  // See https://docs.soliditylang.org/en/latest/control-structures.html#checked-or-unchecked-arithmetic.
                  return prod0 / denominator;
              }
  
              // Make sure the result is less than 2^256. Also prevents denominator == 0.
              if (denominator <= prod1) {
                  revert MathOverflowedMulDiv();
              }
  
              ///////////////////////////////////////////////
              // 512 by 256 division.
              ///////////////////////////////////////////////
  
              // Make division exact by subtracting the remainder from [prod1 prod0].
              uint256 remainder;
              assembly {
                  // Compute remainder using mulmod.
                  remainder := mulmod(x, y, denominator)
  
                  // Subtract 256 bit number from 512 bit number.
                  prod1 := sub(prod1, gt(remainder, prod0))
                  prod0 := sub(prod0, remainder)
              }
  
              // Factor powers of two out of denominator and compute largest power of two divisor of denominator.
              // Always >= 1. See https://cs.stackexchange.com/q/138556/92363.
  
              uint256 twos = denominator & (0 - denominator);
              assembly {
                  // Divide denominator by twos.
                  denominator := div(denominator, twos)
  
                  // Divide [prod1 prod0] by twos.
                  prod0 := div(prod0, twos)
  
                  // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                  twos := add(div(sub(0, twos), twos), 1)
              }
  
              // Shift in bits from prod1 into prod0.
              prod0 |= prod1 * twos;
  
              // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
              // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
              // four bits. That is, denominator * inv = 1 mod 2^4.
              uint256 inverse = (3 * denominator) ^ 2;
  
              // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also
              // works in modular arithmetic, doubling the correct bits in each step.
              inverse *= 2 - denominator * inverse; // inverse mod 2^8
              inverse *= 2 - denominator * inverse; // inverse mod 2^16
              inverse *= 2 - denominator * inverse; // inverse mod 2^32
              inverse *= 2 - denominator * inverse; // inverse mod 2^64
              inverse *= 2 - denominator * inverse; // inverse mod 2^128
              inverse *= 2 - denominator * inverse; // inverse mod 2^256
  
              // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
              // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
              // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
              // is no longer required.
              result = prod0 * inverse;
              return result;
          }
      }
  
      /**
       * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
       */
      function mulDiv(uint256 x, uint256 y, uint256 denominator, Rounding rounding) internal pure returns (uint256) {
          uint256 result = mulDiv(x, y, denominator);
          if (unsignedRoundsUp(rounding) && mulmod(x, y, denominator) > 0) {
              result += 1;
          }
          return result;
      }
  
      /**
       * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded
       * towards zero.
       *
       * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
       */
      function sqrt(uint256 a) internal pure returns (uint256) {
          if (a == 0) {
              return 0;
          }
  
          // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
          //
          // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
          // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
          //
          // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
          // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
          // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
          //
          // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
          uint256 result = 1 << (log2(a) >> 1);
  
          // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
          // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
          // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
          // into the expected uint128 result.
          unchecked {
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              return min(result, a / result);
          }
      }
  
      /**
       * @notice Calculates sqrt(a), following the selected rounding direction.
       */
      function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
          unchecked {
              uint256 result = sqrt(a);
              return result + (unsignedRoundsUp(rounding) && result * result < a ? 1 : 0);
          }
      }
  
      /**
       * @dev Return the log in base 2 of a positive value rounded towards zero.
       * Returns 0 if given 0.
       */
      function log2(uint256 value) internal pure returns (uint256) {
          uint256 result = 0;
          unchecked {
              if (value >> 128 > 0) {
                  value >>= 128;
                  result += 128;
              }
              if (value >> 64 > 0) {
                  value >>= 64;
                  result += 64;
              }
              if (value >> 32 > 0) {
                  value >>= 32;
                  result += 32;
              }
              if (value >> 16 > 0) {
                  value >>= 16;
                  result += 16;
              }
              if (value >> 8 > 0) {
                  value >>= 8;
                  result += 8;
              }
              if (value >> 4 > 0) {
                  value >>= 4;
                  result += 4;
              }
              if (value >> 2 > 0) {
                  value >>= 2;
                  result += 2;
              }
              if (value >> 1 > 0) {
                  result += 1;
              }
          }
          return result;
      }
  
      /**
       * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
       * Returns 0 if given 0.
       */
      function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
          unchecked {
              uint256 result = log2(value);
              return result + (unsignedRoundsUp(rounding) && 1 << result < value ? 1 : 0);
          }
      }
  
      /**
       * @dev Return the log in base 10 of a positive value rounded towards zero.
       * Returns 0 if given 0.
       */
      function log10(uint256 value) internal pure returns (uint256) {
          uint256 result = 0;
          unchecked {
              if (value >= 10 ** 64) {
                  value /= 10 ** 64;
                  result += 64;
              }
              if (value >= 10 ** 32) {
                  value /= 10 ** 32;
                  result += 32;
              }
              if (value >= 10 ** 16) {
                  value /= 10 ** 16;
                  result += 16;
              }
              if (value >= 10 ** 8) {
                  value /= 10 ** 8;
                  result += 8;
              }
              if (value >= 10 ** 4) {
                  value /= 10 ** 4;
                  result += 4;
              }
              if (value >= 10 ** 2) {
                  value /= 10 ** 2;
                  result += 2;
              }
              if (value >= 10 ** 1) {
                  result += 1;
              }
          }
          return result;
      }
  
      /**
       * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
       * Returns 0 if given 0.
       */
      function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
          unchecked {
              uint256 result = log10(value);
              return result + (unsignedRoundsUp(rounding) && 10 ** result < value ? 1 : 0);
          }
      }
  
      /**
       * @dev Return the log in base 256 of a positive value rounded towards zero.
       * Returns 0 if given 0.
       *
       * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
       */
      function log256(uint256 value) internal pure returns (uint256) {
          uint256 result = 0;
          unchecked {
              if (value >> 128 > 0) {
                  value >>= 128;
                  result += 16;
              }
              if (value >> 64 > 0) {
                  value >>= 64;
                  result += 8;
              }
              if (value >> 32 > 0) {
                  value >>= 32;
                  result += 4;
              }
              if (value >> 16 > 0) {
                  value >>= 16;
                  result += 2;
              }
              if (value >> 8 > 0) {
                  result += 1;
              }
          }
          return result;
      }
  
      /**
       * @dev Return the log in base 256, following the selected rounding direction, of a positive value.
       * Returns 0 if given 0.
       */
      function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
          unchecked {
              uint256 result = log256(value);
              return result + (unsignedRoundsUp(rounding) && 1 << (result << 3) < value ? 1 : 0);
          }
      }
  
      /**
       * @dev Returns whether a provided rounding mode is considered rounding up for unsigned integers.
       */
      function unsignedRoundsUp(Rounding rounding) internal pure returns (bool) {
          return uint8(rounding) % 2 == 1;
      }
  }
  
  interface IWeth {
      event Approval(address indexed src, address indexed guy, uint wad);
      event Transfer(address indexed src, address indexed dst, uint wad);
      event Deposit(address indexed dst, uint wad);
      event Withdrawal(address indexed src, uint wad);
  
      function name() external view returns (string memory);
      function symbol() external view returns (string memory);
      function decimals() external view returns (uint8);
      function balanceOf(address user) external view returns (uint256);
      function allowance(address owner, address spender) external view returns (uint256);
      function totalSupply() external view returns (uint);
  
      function deposit() external payable;
      function withdraw(uint256 wad) external;
      function approve(address guy, uint wad) external returns (bool);
      function transfer(address dst, uint wad) external returns (bool);
      function transferFrom(address src, address dst, uint wad) external returns (bool);
  }
  
  interface IstETH {
      function submit(address _referral) external payable returns (uint256);
      function getSharesByPooledEth(uint256 _ethAmount) external view returns (uint256);
      function getPooledEthByShares(uint256 _sharesAmount) external view returns (uint256);
  }
  
  interface IwstETH {
      function wrap(uint256 _stETHAmount) external returns (uint256);
      function unwrap(uint256 _wstETHAmount) external returns (uint256);
      function getWstETHByStETH(uint256 _stETHAmount) external view returns (uint256);
      function getStETHByWstETH(uint256 _wstETHAmount) external view returns (uint256);
      function stEthPerToken() external view returns (uint256);
      function tokensPerStEth() external view returns (uint256);
  }
  
  interface IsDAI {
      function deposit(uint256 assets, address receiver) external returns (uint256 shares);
      function redeem(uint256 shares, address receiver, address owner) external returns (uint256 assets);
      function convertToShares(uint256 assets) external view returns (uint256);
      function convertToAssets(uint256 shares) external view returns (uint256);
  }
  
  interface IeETHLiquidityPool {
      function deposit(address _referral) external payable returns (uint256);
      function sharesForAmount(uint256 _amount) external view returns (uint256);
      function amountForShare(uint256 _share) external view returns (uint256);
  }
  
  interface IweETH {
      function wrap(uint256 _eETHAmount) external returns (uint256);
  }
  
  /**
   * @dev Provides information about the current execution context, including the
   * sender of the transaction and its data. While these are generally available
   * via msg.sender and msg.data, they should not be accessed in such a direct
   * manner, since when dealing with meta-transactions the account sending and
   * paying for execution may not be the actual sender (as far as an application
   * is concerned).
   *
   * This contract is only required for intermediate, library-like contracts.
   */
  abstract contract Context {
      function _msgSender() internal view virtual returns (address) {
          return msg.sender;
      }
  
      function _msgData() internal view virtual returns (bytes calldata) {
          return msg.data;
      }
  
      function _contextSuffixLength() internal view virtual returns (uint256) {
          return 0;
      }
  }
  
  /**
   * @dev Contract module which provides a basic access control mechanism, where
   * there is an account (an owner) that can be granted exclusive access to
   * specific functions.
   *
   * The initial owner is set to the address provided by the deployer. This can
   * later be changed with {transferOwnership}.
   *
   * This module is used through inheritance. It will make available the modifier
   * `onlyOwner`, which can be applied to your functions to restrict their use to
   * the owner.
   */
  abstract contract Ownable is Context {
      address private _owner;
  
      /**
       * @dev The caller account is not authorized to perform an operation.
       */
      error OwnableUnauthorizedAccount(address account);
  
      /**
       * @dev The owner is not a valid owner account. (eg. `address(0)`)
       */
      error OwnableInvalidOwner(address owner);
  
      event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);
  
      /**
       * @dev Initializes the contract setting the address provided by the deployer as the initial owner.
       */
      constructor(address initialOwner) {
          if (initialOwner == address(0)) {
              revert OwnableInvalidOwner(address(0));
          }
          _transferOwnership(initialOwner);
      }
  
      /**
       * @dev Throws if called by any account other than the owner.
       */
      modifier onlyOwner() {
          _checkOwner();
          _;
      }
  
      /**
       * @dev Returns the address of the current owner.
       */
      function owner() public view virtual returns (address) {
          return _owner;
      }
  
      /**
       * @dev Throws if the sender is not the owner.
       */
      function _checkOwner() internal view virtual {
          if (owner() != _msgSender()) {
              revert OwnableUnauthorizedAccount(_msgSender());
          }
      }
  
      /**
       * @dev Leaves the contract without owner. It will not be possible to call
       * `onlyOwner` functions. Can only be called by the current owner.
       *
       * NOTE: Renouncing ownership will leave the contract without an owner,
       * thereby disabling any functionality that is only available to the owner.
       */
      function renounceOwnership() public virtual onlyOwner {
          _transferOwnership(address(0));
      }
  
      /**
       * @dev Transfers ownership of the contract to a new account (`newOwner`).
       * Can only be called by the current owner.
       */
      function transferOwnership(address newOwner) public virtual onlyOwner {
          if (newOwner == address(0)) {
              revert OwnableInvalidOwner(address(0));
          }
          _transferOwnership(newOwner);
      }
  
      /**
       * @dev Transfers ownership of the contract to a new account (`newOwner`).
       * Internal function without access restriction.
       */
      function _transferOwnership(address newOwner) internal virtual {
          address oldOwner = _owner;
          _owner = newOwner;
          emit OwnershipTransferred(oldOwner, newOwner);
      }
  }
  
  /**
   * @dev Contract module which provides access control mechanism, where
   * there is an account (an owner) that can be granted exclusive access to
   * specific functions.
   *
   * The initial owner is specified at deployment time in the constructor for `Ownable`. This
   * can later be changed with {transferOwnership} and {acceptOwnership}.
   *
   * This module is used through inheritance. It will make available all functions
   * from parent (Ownable).
   */
  abstract contract Ownable2Step is Ownable {
      address private _pendingOwner;
  
      event OwnershipTransferStarted(address indexed previousOwner, address indexed newOwner);
  
      /**
       * @dev Returns the address of the pending owner.
       */
      function pendingOwner() public view virtual returns (address) {
          return _pendingOwner;
      }
  
      /**
       * @dev Starts the ownership transfer of the contract to a new account. Replaces the pending transfer if there is one.
       * Can only be called by the current owner.
       */
      function transferOwnership(address newOwner) public virtual override onlyOwner {
          _pendingOwner = newOwner;
          emit OwnershipTransferStarted(owner(), newOwner);
      }
  
      /**
       * @dev Transfers ownership of the contract to a new account (`newOwner`) and deletes any pending owner.
       * Internal function without access restriction.
       */
      function _transferOwnership(address newOwner) internal virtual override {
          delete _pendingOwner;
          super._transferOwnership(newOwner);
      }
  
      /**
       * @dev The new owner accepts the ownership transfer.
       */
      function acceptOwnership() public virtual {
          address sender = _msgSender();
          if (pendingOwner() != sender) {
              revert OwnableUnauthorizedAccount(sender);
          }
          _transferOwnership(sender);
      }
  }
  
  event ReplaceImplementationStarted(address indexed previousImplementation, address indexed newImplementation);
  event ReplaceImplementation(address indexed previousImplementation, address indexed newImplementation);
  error Unauthorized();
  
  contract Upgradeable2Step is Ownable2Step {
      address public pendingImplementation;
      address public implementation;
  
      constructor() Ownable(msg.sender) {}
  
      // called on an inheriting proxy contract
      function replaceImplementation(address impl_) public onlyOwner {
          pendingImplementation = impl_;
          emit ReplaceImplementationStarted(implementation, impl_);
      }
  
      // called from an inheriting implementation contract
      function acceptImplementation() public {
          if (msg.sender != pendingImplementation) {
              revert OwnableUnauthorizedAccount(msg.sender);
          }
          emit ReplaceImplementation(implementation, msg.sender);
          delete pendingImplementation;
          implementation = msg.sender;
      }
  
      // called on an inheriting implementation contract
      function becomeImplementation(Upgradeable2Step proxy) public {
          if (msg.sender != proxy.owner()) {
              revert Unauthorized();
          }
          proxy.acceptImplementation();
      }
  }
  
  interface BridgeLike {
      function deposit(
          address _l2Receiver,
          address _l1Token,
          uint256 _amount,
          uint256 _l2TxGasLimit,
          uint256 _l2TxGasPerPubdataByte,
          address _refundRecipient
      ) external payable returns (bytes32 l2TxHash);
  }
  
  contract SophonFarmingState {
  
      // Info of each pool.
      struct PoolInfo {
          IERC20 lpToken; // Address of LP token contract.
          address l2Farm; // Address of the farming contract on Sophon chain
          uint256 amount; // total amount of LP tokens earning yield from deposits and boosts
          uint256 boostAmount; // total boosted value purchased by users
          uint256 depositAmount; // remaining deposits not applied to a boost purchases
          uint256 allocPoint; // How many allocation points assigned to this pool. Points to distribute per block.
          uint256 lastRewardBlock; // Last block number that points distribution occurs.
          uint256 accPointsPerShare; // Accumulated points per share, times 1e18. See below.
          string description; // Description of pool.
      }
  
      // Info of each user.
      struct UserInfo {
          uint256 amount; // Amount of LP tokens the user is earning yield on from deposits and boosts
          uint256 boostAmount; // Boosted value purchased by the user
          uint256 depositAmount; // remaining deposits not applied to a boost purchases
          uint256 rewardSettled; // rewards settled
          uint256 rewardDebt; // rewards debt
      }
  
      enum PredefinedPool {
          sDAI,          // MakerDAO (sDAI)
          wstETH,        // Lido (wstETH)
          weETH          // ether.fi (weETH)
      }
  
      mapping(PredefinedPool => uint256) public typeToId;
  
      // held proceeds from booster sales
      mapping(uint256 => uint256) public heldProceeds;
  
      uint256 public boosterMultiplier;
  
      // Points created per block.
      uint256 public pointsPerBlock;
  
      // Info of each pool.
      PoolInfo[] public poolInfo;
  
      // Info of each user that stakes LP tokens.
      mapping(uint256 => mapping(address => UserInfo)) public userInfo;
  
      // Total allocation points. Must be the sum of all allocation points in all pools.
      uint256 public totalAllocPoint;
  
      // The block number when point mining ends.
      uint256 public endBlock;
  
      bool internal _initialized;
  
      mapping(address => bool) public poolExists;
  
      uint256 public endBlockForWithdrawals;
  
      BridgeLike public bridge;
      mapping(uint256 => bool) public isBridged;
  
      mapping(address userAdmin => mapping(address user => bool inWhitelist)) public whitelist;
  }
  
  /**
   * @title Sophon Farming Contract
   * @author Sophon
   */
  contract SophonFarming is Upgradeable2Step, SophonFarmingState {
      using SafeERC20 for IERC20;
  
      /// @notice Emitted when a new pool is added
      event Add(address indexed lpToken, uint256 indexed pid, uint256 allocPoint);
  
      /// @notice Emitted when a pool is updated
      event Set(address indexed lpToken, uint256 indexed pid, uint256 allocPoint);
  
      /// @notice Emitted when a user deposits to a pool
      event Deposit(address indexed user, uint256 indexed pid, uint256 depositAmount, uint256 boostAmount);
  
      /// @notice Emitted when a user withdraws from a pool
      event Withdraw(address indexed user, uint256 indexed pid, uint256 amount);
  
      /// @notice Emitted when a whitelisted admin transfers points from one user to another
      event TransferPoints(address indexed sender, address indexed receiver, uint256 indexed pid, uint256 amount);
  
      /// @notice Emitted when a user increases the boost of an existing deposit
      event IncreaseBoost(address indexed user, uint256 indexed pid, uint256 boostAmount);
  
      /// @notice Emitted when all pool funds are bridged to Sophon blockchain
      event BridgePool(address indexed user, uint256 indexed pid, uint256 amount);
  
      /// @notice Emitted when the admin bridges booster proceeds
      event BridgeProceeds(uint256 indexed pid, uint256 proceeds);
  
      /// @notice Emitted when the the revertFailedBridge function is called
      event RevertFailedBridge(uint256 indexed pid);
  
      /// @notice Emitted when the the updatePool function is called
      event PoolUpdated(uint256 indexed pid);
  
      /// @notice Emitted when setPointsPerBlock is called
      event SetPointsPerBlock(uint256 oldValue, uint256 newValue);
  
      error ZeroAddress();
      error PoolExists();
      error PoolDoesNotExist();
      error AlreadyInitialized();
      error NotFound(address lpToken);
      error FarmingIsStarted();
      error FarmingIsEnded();
      error TransferNotAllowed();
      error TransferTooHigh(uint256 maxAllowed);
      error InvalidEndBlock();
      error InvalidDeposit();
      error InvalidBooster();
      error InvalidPointsPerBlock();
      error InvalidTransfer();
      error WithdrawNotAllowed();
      error WithdrawTooHigh(uint256 maxAllowed);
      error WithdrawIsZero();
      error NothingInPool();
      error NoEthSent();
      error BoostTooHigh(uint256 maxAllowed);
      error BoostIsZero();
      error BridgeInvalid();
  
      address public immutable dai;
      address public immutable sDAI;
      address public immutable weth;
      address public immutable stETH;
      address public immutable wstETH;
      address public immutable eETH;
      address public immutable eETHLiquidityPool;
      address public immutable weETH;
  
      /**
       * @notice Construct SophonFarming
       * @param tokens_ Immutable token addresses
       * @dev 0:dai, 1:sDAI, 2:weth, 3:stETH, 4:wstETH, 5:eETH, 6:eETHLiquidityPool, 7:weETH
       */
      constructor(address[8] memory tokens_) {
          dai = tokens_[0];
          sDAI = tokens_[1];
          weth = tokens_[2];
          stETH = tokens_[3];
          wstETH = tokens_[4];
          eETH = tokens_[5];
          eETHLiquidityPool = tokens_[6];
          weETH = tokens_[7];
      }
  
      /**
       * @notice Allows direct deposits of ETH for deposit to the wstETH pool
       */
      receive() external payable {
          if (msg.sender == weth) {
              return;
          }
  
          depositEth(0, PredefinedPool.wstETH);
      }
  
      /**
       * @notice Initialize the farm
       * @param wstEthAllocPoint_ wstEth alloc points
       * @param weEthAllocPoint_ weEth alloc points
       * @param sDAIAllocPoint_ sdai alloc points
       * @param _pointsPerBlock points per block
       * @param _initialPoolStartBlock start block
       * @param _boosterMultiplier booster multiplier
       */
      function initialize(uint256 wstEthAllocPoint_, uint256 weEthAllocPoint_, uint256 sDAIAllocPoint_, uint256 _pointsPerBlock, uint256 _initialPoolStartBlock, uint256 _boosterMultiplier) public virtual onlyOwner {
          if (_initialized) {
              revert AlreadyInitialized();
          }
  
          if (_pointsPerBlock < 1e18 || _pointsPerBlock > 1000e18) {
              revert InvalidPointsPerBlock();
          }
          pointsPerBlock = _pointsPerBlock;
  
          if (_boosterMultiplier < 1e18 || _boosterMultiplier > 10e18) {
              revert InvalidBooster();
          }
          boosterMultiplier = _boosterMultiplier;
  
          poolExists[dai] = true;
          poolExists[weth] = true;
          poolExists[stETH] = true;
          poolExists[eETH] = true;
  
          _initialized = true;
  
          // sDAI
          typeToId[PredefinedPool.sDAI] = add(sDAIAllocPoint_, sDAI, "sDAI", _initialPoolStartBlock, 0);
          IERC20(dai).approve(sDAI, 2**256-1);
  
          // wstETH
          typeToId[PredefinedPool.wstETH] = add(wstEthAllocPoint_, wstETH, "wstETH", _initialPoolStartBlock, 0);
          IERC20(stETH).approve(wstETH, 2**256-1);
  
          // weETH
          typeToId[PredefinedPool.weETH] = add(weEthAllocPoint_, weETH, "weETH", _initialPoolStartBlock, 0);
          IERC20(eETH).approve(weETH, 2**256-1);
      }
  
      /**
       * @notice Adds a new pool to the farm. Can only be called by the owner.
       * @param _allocPoint alloc point for new pool
       * @param _lpToken lpToken address
       * @param _description description of new pool
       * @param _poolStartBlock block at which points start to accrue for the pool
       * @param _newPointsPerBlock update global points per block; 0 means no update
       * @return uint256 The pid of the newly created asset
       */
      function add(uint256 _allocPoint, address _lpToken, string memory _description, uint256 _poolStartBlock, uint256 _newPointsPerBlock) public onlyOwner returns (uint256) {
          if (_lpToken == address(0)) {
              revert ZeroAddress();
          }
          if (poolExists[_lpToken]) {
              revert PoolExists();
          }
          if (isFarmingEnded()) {
              revert FarmingIsEnded();
          }
  
          if (_newPointsPerBlock != 0) {
              setPointsPerBlock(_newPointsPerBlock);
          } else {
              massUpdatePools();
          }
  
          uint256 lastRewardBlock =
              getBlockNumber() > _poolStartBlock ? getBlockNumber() : _poolStartBlock;
          totalAllocPoint = totalAllocPoint + _allocPoint;
          poolExists[_lpToken] = true;
  
          uint256 pid = poolInfo.length;
  
          poolInfo.push(
              PoolInfo({
                  lpToken: IERC20(_lpToken),
                  l2Farm: address(0),
                  amount: 0,
                  boostAmount: 0,
                  depositAmount: 0,
                  allocPoint: _allocPoint,
                  lastRewardBlock: lastRewardBlock,
                  accPointsPerShare: 0,
                  description: _description
              })
          );
  
          emit Add(_lpToken, pid, _allocPoint);
  
          return pid;
      }
  
      /**
       * @notice Updates the given pool's allocation point. Can only be called by the owner.
       * @param _pid The pid to update
       * @param _allocPoint The new alloc point to set for the pool
       * @param _poolStartBlock block at which points start to accrue for the pool; 0 means no update
       * @param _newPointsPerBlock update global points per block; 0 means no update
       */
      function set(uint256 _pid, uint256 _allocPoint, uint256 _poolStartBlock, uint256 _newPointsPerBlock) external onlyOwner {
          if (isFarmingEnded()) {
              revert FarmingIsEnded();
          }
  
          if (_newPointsPerBlock != 0) {
              setPointsPerBlock(_newPointsPerBlock);
          } else {
              massUpdatePools();
          }
  
          PoolInfo storage pool = poolInfo[_pid];
          address lpToken = address(pool.lpToken);
          if (lpToken == address(0) || !poolExists[lpToken]) {
              revert PoolDoesNotExist();
          }
          totalAllocPoint = totalAllocPoint - pool.allocPoint + _allocPoint;
          pool.allocPoint = _allocPoint;
  
          // pool starting block is updated if farming hasn't started and _poolStartBlock is non-zero
          if (_poolStartBlock != 0 && getBlockNumber() < pool.lastRewardBlock) {
              pool.lastRewardBlock =
                  getBlockNumber() > _poolStartBlock ? getBlockNumber() : _poolStartBlock;
          }
  
          emit Set(lpToken, _pid, _allocPoint);
      }
  
      /**
       * @notice Returns the number of pools in the farm
       * @return uint256 number of pools
       */
      function poolLength() external view returns (uint256) {
          return poolInfo.length;
      }
  
      /**
       * @notice Checks if farming is ended
       * @return bool True if farming is ended
       */
      function isFarmingEnded() public view returns (bool) {
          uint256 _endBlock = endBlock;
          if (_endBlock != 0 && getBlockNumber() > _endBlock) {
              return true;
          } else {
              return false;
          }
      }
  
      /**
       * @notice Checks if the withdrawal period is ended
       * @return bool True if withdrawal period is ended
       */
      function isWithdrawPeriodEnded() public view returns (bool) {
          uint256 _endBlockForWithdrawals = endBlockForWithdrawals;
          if (_endBlockForWithdrawals != 0 && getBlockNumber() > _endBlockForWithdrawals) {
              return true;
          } else {
              return false;
          }
      }
  
      /**
       * @notice Updates the bridge contract
       */
      function setBridge(address _bridge) external onlyOwner {
          if (_bridge == address(0)) {
              revert ZeroAddress();
          }
          bridge = BridgeLike(_bridge);
      }
  
      /**
       * @notice Updates the L2 Farm for the pool
       * @param _pid the pid
       * @param _l2Farm the l2Farm address
       */
      function setL2FarmForPool(uint256 _pid, address _l2Farm) external onlyOwner {
          if (_l2Farm == address(0)) {
              revert ZeroAddress();
          }
          if (address(poolInfo[_pid].lpToken) == address(0)) {
              revert PoolDoesNotExist();
          }
          poolInfo[_pid].l2Farm = _l2Farm;
      }
  
      /**
       * @notice Set the end block of the farm
       * @param _endBlock the end block
       * @param _withdrawalBlocks the last block that withdrawals are allowed
       */
      function setEndBlock(uint256 _endBlock, uint256 _withdrawalBlocks) external onlyOwner {
          if (isFarmingEnded()) {
              revert FarmingIsEnded();
          }
          uint256 _endBlockForWithdrawals;
          if (_endBlock != 0) {
              if (getBlockNumber() > _endBlock) {
                  revert InvalidEndBlock();
              }
              _endBlockForWithdrawals = _endBlock + _withdrawalBlocks;
          } else {
              // withdrawal blocks needs an endBlock
              _endBlockForWithdrawals = 0;
          }
          massUpdatePools();
          endBlock = _endBlock;
          endBlockForWithdrawals = _endBlockForWithdrawals;
      }
  
      /**
       * @notice Set points per block
       * @param _pointsPerBlock points per block to set
       */
      function setPointsPerBlock(uint256 _pointsPerBlock) public onlyOwner {
          if (isFarmingEnded()) {
              revert FarmingIsEnded();
          }
          if (_pointsPerBlock < 1e18 || _pointsPerBlock > 1000e18) {
              revert InvalidPointsPerBlock();
          }
          massUpdatePools();
          emit SetPointsPerBlock(pointsPerBlock, _pointsPerBlock);
          pointsPerBlock = _pointsPerBlock;
      }
  
      /**
       * @notice Set booster multiplier
       * @param _boosterMultiplier booster multiplier to set
       */
      function setBoosterMultiplier(uint256 _boosterMultiplier) external onlyOwner {
          if (_boosterMultiplier < 1e18 || _boosterMultiplier > 10e18) {
              revert InvalidBooster();
          }
          if (isFarmingEnded()) {
              revert FarmingIsEnded();
          }
          massUpdatePools();
          boosterMultiplier = _boosterMultiplier;
      }
  
      /**
       * @notice Returns the block multiplier
       * @param _from from block
       * @param _to to block
       * @return uint256 The block multiplier
       */
      function _getBlockMultiplier(uint256 _from, uint256 _to) internal view returns (uint256) {
          uint256 _endBlock = endBlock;
          if (_endBlock != 0) {
              _to = Math.min(_to, _endBlock);
          }
          if (_to > _from) {
              return (_to - _from) * 1e18;
          } else {
              return 0;
          }
      }
  
      /**
       * @notice Adds or removes users from the whitelist
       * @param _userAdmin an admin user who can transfer points for users
       * @param _users list of users
       * @param _isInWhitelist to add or remove
       */
      function setUsersWhitelisted(address _userAdmin, address[] memory _users, bool _isInWhitelist) external onlyOwner {
          for(uint i = 0; i < _users.length; i++) {
              whitelist[_userAdmin][_users[i]] = _isInWhitelist;
          }
      }
  
      /**
       * @notice Returns pending points for user in a pool
       * @param _pid pid of the pool
       * @param _user user in the pool
       * @return uint256 pendings points
       */
      function _pendingPoints(uint256 _pid, address _user) internal view returns (uint256) {
          PoolInfo storage pool = poolInfo[_pid];
          UserInfo storage user = userInfo[_pid][_user];
  
          uint256 accPointsPerShare = pool.accPointsPerShare;
  
          uint256 lpSupply = pool.amount;
          if (getBlockNumber() > pool.lastRewardBlock && lpSupply != 0) {
              uint256 blockMultiplier = _getBlockMultiplier(pool.lastRewardBlock, getBlockNumber());
  
              uint256 pointReward =
                  blockMultiplier *
                  pointsPerBlock *
                  pool.allocPoint /
                  totalAllocPoint;
  
              accPointsPerShare = pointReward /
                  lpSupply +
                  accPointsPerShare;
          }
  
          return user.amount *
              accPointsPerShare /
              1e18 +
              user.rewardSettled -
              user.rewardDebt;
      }
  
      /**
       * @notice Returns pending points for user in a pool
       * @param _pid pid of the pool
       * @param _user user in the pool
       * @return uint256 pendings points
       */
      function pendingPoints(uint256 _pid, address _user) external view returns (uint256) {
          return _pendingPoints(_pid, _user);
      }
  
      /**
       * @notice Update accounting of all pools
       */
      function massUpdatePools() public {
          uint256 length = poolInfo.length;
          for(uint256 pid = 0; pid < length; ++pid) {
              updatePool(pid);
          }
      }
  
      /**
       * @notice Updating accounting of a single pool
       * @param _pid pid to update
       */
      function updatePool(uint256 _pid) public {
          PoolInfo storage pool = poolInfo[_pid];
          if (getBlockNumber() <= pool.lastRewardBlock) {
              return;
          }
          uint256 lpSupply = pool.amount;
          uint256 _pointsPerBlock = pointsPerBlock;
          uint256 _allocPoint = pool.allocPoint;
          if (lpSupply == 0 || _pointsPerBlock == 0 || _allocPoint == 0) {
              pool.lastRewardBlock = getBlockNumber();
              return;
          }
          uint256 blockMultiplier = _getBlockMultiplier(pool.lastRewardBlock, getBlockNumber());
          uint256 pointReward =
              blockMultiplier *
              _pointsPerBlock *
              _allocPoint /
              totalAllocPoint;
  
          pool.accPointsPerShare = pointReward /
              lpSupply +
              pool.accPointsPerShare;
  
          pool.lastRewardBlock = getBlockNumber();
  
          emit PoolUpdated(_pid);
      }
  
      /**
       * @notice Deposit assets to SophonFarming
       * @param _pid pid of the pool
       * @param _amount amount of the deposit
       * @param _boostAmount amount to boost
       */
      function deposit(uint256 _pid, uint256 _amount, uint256 _boostAmount) external {
          poolInfo[_pid].lpToken.safeTransferFrom(
              msg.sender,
              address(this),
              _amount
          );
  
          _deposit(_pid, _amount, _boostAmount);
      }
  
      /**
       * @notice Deposit DAI to SophonFarming
       * @param _amount amount of the deposit
       * @param _boostAmount amount to boost
       */
      function depositDai(uint256 _amount, uint256 _boostAmount) external {
          IERC20(dai).safeTransferFrom(
              msg.sender,
              address(this),
              _amount
          );
  
          _depositPredefinedAsset(_amount, _amount, _boostAmount, PredefinedPool.sDAI);
      }
  
      /**
       * @notice Deposit stETH to SophonFarming
       * @param _amount amount of the deposit
       * @param _boostAmount amount to boost
       */
      function depositStEth(uint256 _amount, uint256 _boostAmount) external {
          uint256 beforeBalance = IERC20(stETH).balanceOf(address(this));
          IERC20(stETH).safeTransferFrom(
              msg.sender,
              address(this),
              _amount
          );
          uint256 _finalAmount = IERC20(stETH).balanceOf(address(this)) - beforeBalance;
  
          _depositPredefinedAsset(_finalAmount, _amount, _boostAmount, PredefinedPool.wstETH);
      }
  
      /**
       * @notice Deposit eETH to SophonFarming
       * @param _amount amount of the deposit
       * @param _boostAmount amount to boost
       */
      function depositeEth(uint256 _amount, uint256 _boostAmount) external {
          uint256 beforeBalance = IERC20(eETH).balanceOf(address(this));
          IERC20(eETH).safeTransferFrom(
              msg.sender,
              address(this),
              _amount
          );
          uint256 _finalAmount = IERC20(eETH).balanceOf(address(this)) - beforeBalance;
  
          _depositPredefinedAsset(_finalAmount, _amount, _boostAmount, PredefinedPool.weETH);
      }
  
      /**
       * @notice Deposit ETH to SophonFarming when specifying a pool
       * @param _boostAmount amount to boost
       * @param _predefinedPool specific pool type to deposit to
       */
      function depositEth(uint256 _boostAmount, PredefinedPool _predefinedPool) public payable {
          if (msg.value == 0) {
              revert NoEthSent();
          }
  
          uint256 _finalAmount = msg.value;
          if (_predefinedPool == PredefinedPool.wstETH) {
              _finalAmount = _ethTOstEth(_finalAmount);
          } else if (_predefinedPool == PredefinedPool.weETH) {
              _finalAmount = _ethTOeEth(_finalAmount);
          } else {
              revert InvalidDeposit();
          }
  
          _depositPredefinedAsset(_finalAmount, msg.value, _boostAmount, _predefinedPool);
      }
  
      /**
       * @notice Deposit WETH to SophonFarming when specifying a pool
       * @param _amount amount of the deposit
       * @param _boostAmount amount to boost
       * @param _predefinedPool specific pool type to deposit to
       */
      function depositWeth(uint256 _amount, uint256 _boostAmount, PredefinedPool _predefinedPool) external {
          IERC20(weth).safeTransferFrom(
              msg.sender,
              address(this),
              _amount
          );
  
          uint256 _finalAmount = _wethTOEth(_amount);
          if (_predefinedPool == PredefinedPool.wstETH) {
              _finalAmount = _ethTOstEth(_finalAmount);
          } else if (_predefinedPool == PredefinedPool.weETH) {
              _finalAmount = _ethTOeEth(_finalAmount);
          } else {
              revert InvalidDeposit();
          }
  
          _depositPredefinedAsset(_finalAmount, _amount, _boostAmount, _predefinedPool);
      }
  
      /**
       * @notice Deposit a predefined asset to SophonFarming
       * @param _amount amount of the deposit
       * @param _initalAmount amount of the deposit prior to conversions
       * @param _boostAmount amount to boost
       * @param _predefinedPool specific pool type to deposit to
       */
      function _depositPredefinedAsset(uint256 _amount, uint256 _initalAmount, uint256 _boostAmount, PredefinedPool _predefinedPool) internal {
  
          uint256 _finalAmount;
  
          if (_predefinedPool == PredefinedPool.sDAI) {
              _finalAmount = _daiTOsDai(_amount);
          } else if (_predefinedPool == PredefinedPool.wstETH) {
              _finalAmount = _stEthTOwstEth(_amount);
          } else if (_predefinedPool == PredefinedPool.weETH) {
              _finalAmount = _eethTOweEth(_amount);
          } else {
              revert InvalidDeposit();
          }
  
          // adjust boostAmount for the new asset
          _boostAmount = _boostAmount * _finalAmount / _initalAmount;
  
          _deposit(typeToId[_predefinedPool], _finalAmount, _boostAmount);
      }
  
      /**
       * @notice Deposit an asset to SophonFarming
       * @param _pid pid of the deposit
       * @param _depositAmount amount of the deposit
       * @param _boostAmount amount to boost
       */
      function _deposit(uint256 _pid, uint256 _depositAmount, uint256 _boostAmount) internal {
          if (isFarmingEnded()) {
              revert FarmingIsEnded();
          }
          if (_depositAmount == 0) {
              revert InvalidDeposit();
          }
          if (_boostAmount > _depositAmount) {
              revert BoostTooHigh(_depositAmount);
          }
  
          PoolInfo storage pool = poolInfo[_pid];
          UserInfo storage user = userInfo[_pid][msg.sender];
          updatePool(_pid);
  
          uint256 userAmount = user.amount;
  
          user.rewardSettled =
              userAmount *
              pool.accPointsPerShare /
              1e18 +
              user.rewardSettled -
              user.rewardDebt;
  
          // booster purchase proceeds
          heldProceeds[_pid] = heldProceeds[_pid] + _boostAmount;
  
          // deposit amount is reduced by amount of the deposit to boost
          _depositAmount = _depositAmount - _boostAmount;
  
          // set deposit amount
          user.depositAmount = user.depositAmount + _depositAmount;
          pool.depositAmount = pool.depositAmount + _depositAmount;
  
          // apply the boost multiplier
          _boostAmount = _boostAmount * boosterMultiplier / 1e18;
  
          user.boostAmount = user.boostAmount + _boostAmount;
          pool.boostAmount = pool.boostAmount + _boostAmount;
  
          // userAmount is increased by remaining deposit amount + full boosted amount
          userAmount = userAmount + _depositAmount + _boostAmount;
  
          user.amount = userAmount;
          pool.amount = pool.amount + _depositAmount + _boostAmount;
  
          user.rewardDebt = userAmount *
              pool.accPointsPerShare /
              1e18;
  
          emit Deposit(msg.sender, _pid, _depositAmount, _boostAmount);
      }
  
      /**
       * @notice Increase boost from existing deposits
       * @param _pid pid to pool
       * @param _boostAmount amount to boost
       */
      function increaseBoost(uint256 _pid, uint256 _boostAmount) external {
          if (isFarmingEnded()) {
              revert FarmingIsEnded();
          }
  
          if (_boostAmount == 0) {
              revert BoostIsZero();
          }
  
          uint256 maxAdditionalBoost = getMaxAdditionalBoost(msg.sender, _pid);
          if (_boostAmount > maxAdditionalBoost) {
              revert BoostTooHigh(maxAdditionalBoost);
          }
  
          PoolInfo storage pool = poolInfo[_pid];
          UserInfo storage user = userInfo[_pid][msg.sender];
          updatePool(_pid);
  
          uint256 userAmount = user.amount;
  
          user.rewardSettled =
              userAmount *
              pool.accPointsPerShare /
              1e18 +
              user.rewardSettled -
              user.rewardDebt;
  
          // booster purchase proceeds
          heldProceeds[_pid] = heldProceeds[_pid] + _boostAmount;
  
          // user's remaining deposit is reduced by amount of the deposit to boost
          user.depositAmount = user.depositAmount - _boostAmount;
          pool.depositAmount = pool.depositAmount - _boostAmount;
  
          // apply the multiplier
          uint256 finalBoostAmount = _boostAmount * boosterMultiplier / 1e18;
  
          user.boostAmount = user.boostAmount + finalBoostAmount;
          pool.boostAmount = pool.boostAmount + finalBoostAmount;
  
          // user amount is increased by the full boosted amount - deposit amount used to boost
          userAmount = userAmount + finalBoostAmount - _boostAmount;
  
          user.amount = userAmount;
          pool.amount = pool.amount + finalBoostAmount - _boostAmount;
  
          user.rewardDebt = userAmount *
              pool.accPointsPerShare /
              1e18;
  
          emit IncreaseBoost(msg.sender, _pid, finalBoostAmount);
      }
  
      /**
       * @notice Returns max additional boost amount allowed to boost current deposits
       * @dev total allowed boost is 100% of total deposit
       * @param _user user in pool
       * @param _pid pid of pool
       * @return uint256 max additional boost
       */
      function getMaxAdditionalBoost(address _user, uint256 _pid) public view returns (uint256) {
          return userInfo[_pid][_user].depositAmount;
      }
  
      /**
       * @notice Withdraw an asset to SophonFarming
       * @param _pid pid of the withdraw
       * @param _withdrawAmount amount of the withdraw
       */
      function withdraw(uint256 _pid, uint256 _withdrawAmount) external {
          if (isWithdrawPeriodEnded()) {
              revert WithdrawNotAllowed();
          }
          if (_withdrawAmount == 0) {
              revert WithdrawIsZero();
          }
  
          PoolInfo storage pool = poolInfo[_pid];
          UserInfo storage user = userInfo[_pid][msg.sender];
          updatePool(_pid);
  
          uint256 userDepositAmount = user.depositAmount;
  
          if (_withdrawAmount == type(uint256).max) {
              _withdrawAmount = userDepositAmount;
          } else if (_withdrawAmount > userDepositAmount) {
              revert WithdrawTooHigh(userDepositAmount);
          }
  
          uint256 userAmount = user.amount;
  
          user.rewardSettled =
              userAmount *
              pool.accPointsPerShare /
              1e18 +
              user.rewardSettled -
              user.rewardDebt;
  
          user.depositAmount = userDepositAmount - _withdrawAmount;
          pool.depositAmount = pool.depositAmount - _withdrawAmount;
  
          userAmount = userAmount - _withdrawAmount;
  
          user.amount = userAmount;
          pool.amount = pool.amount - _withdrawAmount;
  
          user.rewardDebt = userAmount *
              pool.accPointsPerShare /
              1e18;
  
          pool.lpToken.safeTransfer(msg.sender, _withdrawAmount);
  
          emit Withdraw(msg.sender, _pid, _withdrawAmount);
      }
  
      /**
       * @notice Permissionless function to allow anyone to bridge during the correct period
       * @param _pid pid to bridge
       */
      function bridgePool(uint256 _pid, uint256 _l2TxGasLimit, uint256 _l2TxGasPerPubdataByte) external payable {
          revert Unauthorized(); // NOTE: function not fully implemented, an upgrade will implement this later
  
          if (!isFarmingEnded() || !isWithdrawPeriodEnded() || isBridged[_pid]) {
              revert Unauthorized();
          }
  
          updatePool(_pid);
          PoolInfo storage pool = poolInfo[_pid];
  
          uint256 depositAmount = pool.depositAmount;
          if (depositAmount == 0 || address(bridge) == address(0) || pool.l2Farm == address(0)) {
              revert BridgeInvalid();
          }
  
          isBridged[_pid] = true;
  
          IERC20 lpToken = pool.lpToken;
          lpToken.approve(address(bridge), depositAmount);
  
          // Actual values are pending the launch of Sophon testnet
          bridge.deposit{value: msg.value}(
              pool.l2Farm,            // _l2Receiver
              address(lpToken),       // _l1Token
              depositAmount,          // _amount
              _l2TxGasLimit,          // _l2TxGasLimit
              _l2TxGasPerPubdataByte, // _l2TxGasPerPubdataByte
              owner()                 // _refundRecipient
          );
  
          emit BridgePool(msg.sender, _pid, depositAmount);
      }
  
      // TODO: does this function need to call claimFailedDeposit on the bridge?
      // This is pending the launch of Sophon testnet
      /**
       * @notice Called by an admin if a bridge process to Sophon fails
       * @param _pid pid of the failed bridge to revert
       */
      function revertFailedBridge(uint256 _pid) external onlyOwner {
          revert Unauthorized(); // NOTE: function not fully implemented, an upgrade will implement this later
  
          if (address(poolInfo[_pid].lpToken) == address(0)) {
              revert PoolDoesNotExist();
          }
          isBridged[_pid] = false;
          emit RevertFailedBridge(_pid);
      }
  
      /**
       * @notice Called by an whitelisted admin to transfer points to another user
       * @param _pid pid of the pool to transfer points from
       * @param _sender address to send accrued points
       * @param _receiver address to receive accrued points
       * @param _transferAmount amount of points to transfer
       */
      function transferPoints(uint256 _pid, address _sender, address _receiver, uint256 _transferAmount) external {
  
          if (!whitelist[msg.sender][_sender]) {
              revert TransferNotAllowed();
          }
  
          if (_sender == _receiver || _receiver == address(this) || _transferAmount == 0) {
              revert InvalidTransfer();
          }
  
          PoolInfo storage pool = poolInfo[_pid];
  
          if (address(pool.lpToken) == address(0)) {
              revert PoolDoesNotExist();
          }
  
          updatePool(_pid);
          uint256 accPointsPerShare = pool.accPointsPerShare;
  
          UserInfo storage userFrom = userInfo[_pid][_sender];
          UserInfo storage userTo = userInfo[_pid][_receiver];
  
          uint256 userFromAmount = userFrom.amount;
          uint256 userToAmount = userTo.amount;
  
          uint userFromRewardSettled =
              userFromAmount *
              accPointsPerShare /
              1e18 +
              userFrom.rewardSettled -
              userFrom.rewardDebt;
  
          if (_transferAmount == type(uint256).max) {
              _transferAmount = userFromRewardSettled;
          } else if (_transferAmount > userFromRewardSettled) {
              revert TransferTooHigh(userFromRewardSettled);
          }
  
          userFrom.rewardSettled = userFromRewardSettled - _transferAmount;
  
          userTo.rewardSettled =
              userToAmount *
              accPointsPerShare /
              1e18 +
              userTo.rewardSettled -
              userTo.rewardDebt +
              _transferAmount;
  
          userFrom.rewardDebt = userFromAmount *
              accPointsPerShare /
              1e18;
  
          userTo.rewardDebt = userToAmount *
              accPointsPerShare /
              1e18;
  
          emit TransferPoints(_sender, _receiver, _pid, _transferAmount);
      }
  
      /**
       * @notice Converts WETH to ETH
       * @dev WETH withdrawl
       * @param _amount in amount
       * @return uint256 out amount
       */
      function _wethTOEth(uint256 _amount) internal returns (uint256) {
          // unwrap weth to eth
          IWeth(weth).withdraw(_amount);
          return _amount;
      }
  
      /**
       * @notice Converts ETH to stETH
       * @dev Lido
       * @param _amount in amount
       * @return uint256 out amount
       */
      function _ethTOstEth(uint256 _amount) internal returns (uint256) {
          return IstETH(stETH).getPooledEthByShares(
              IstETH(stETH).submit{value: _amount}(owner())
          );
      }
  
      /**
       * @notice Converts stETH to wstETH
       * @dev Lido
       * @param _amount in amount
       * @return uint256 out amount
       */
      function _stEthTOwstEth(uint256 _amount) internal returns (uint256) {
          // wrap returns exact amount of wstETH
          return IwstETH(wstETH).wrap(_amount);
      }
  
      /**
       * @notice Converts ETH to eETH
       * @dev ether.fi
       * @param _amount in amount
       * @return uint256 out amount
       */
      function _ethTOeEth(uint256 _amount) internal returns (uint256) {
          return IeETHLiquidityPool(eETHLiquidityPool).amountForShare(
              IeETHLiquidityPool(eETHLiquidityPool).deposit{value: _amount}(owner())
          );
      }
  
      /**
       * @notice Converts eETH to weETH
       * @dev ether.fi
       * @param _amount in amount
       * @return uint256 out amount
       */
      function _eethTOweEth(uint256 _amount) internal returns (uint256) {
          // wrap returns exact amount of weETH
          return IweETH(weETH).wrap(_amount);
      }
  
      /**
       * @notice Converts DAI to sDAI
       * @dev MakerDao
       * @param _amount in amount
       * @return uint256 out amount
       */
      function _daiTOsDai(uint256 _amount) internal returns (uint256) {
          // deposit DAI to sDAI
          return IsDAI(sDAI).deposit(_amount, address(this));
      }
  
      // This is pending the launch of Sophon testnet
      /**
       * @notice Allows an admin to bridge booster proceeds
       * @param _pid pid to bridge proceeds from
       */
      function bridgeProceeds(uint256 _pid, uint256 _l2TxGasLimit, uint256 _l2TxGasPerPubdataByte) external payable onlyOwner {
          revert Unauthorized(); // NOTE: function not fully implemented, an upgrade will implement this later
  
          uint256 _proceeds = heldProceeds[_pid];
          heldProceeds[_pid] = 0;
  
          // TODO: add bridging logic
  
          emit BridgeProceeds(_pid, _proceeds);
      }
  
      /**
       * @notice Returns the current block number
       * @dev Included to help with testing since it can be overridden for custom functionality
       * @return uint256 current block number
       */
      function getBlockNumber() virtual public view returns (uint256) {
          return block.number;
      }
  
      /**
       * @notice Returns info about each pool
       * @return poolInfos all pool info
       */
      function getPoolInfo() external view returns (PoolInfo[] memory) {
          return poolInfo;
      }
  
      /**
       * @notice Returns user info for a list of users
       * @param _users list of users
       * @return userInfos optimized user info
       */
      function getOptimizedUserInfo(address[] memory _users) external view returns (uint256[4][][] memory userInfos) {
          uint256 usersLen = _users.length;
          userInfos = new uint256[4][][](usersLen);
          uint256 poolLen = poolInfo.length;
          for(uint256 i = 0; i < usersLen; i++) {
              address _user = _users[i];
              userInfos[i] = new uint256[4][](poolLen);
              for(uint256 pid = 0; pid < poolLen; ++pid) {
                  UserInfo memory uinfo = userInfo[pid][_user];
                  userInfos[i][pid][0] = uinfo.amount;
                  userInfos[i][pid][1] = uinfo.boostAmount;
                  userInfos[i][pid][2] = uinfo.depositAmount;
                  userInfos[i][pid][3] = _pendingPoints(pid, _user);
              }
          }
      }
  
      /**
       * @notice Returns accrued points for a list of users
       * @param _users list of users
       * @return pendings accured points for user
       */
      function getPendingPoints(address[] memory _users) external view returns (uint256[][] memory pendings) {
          uint256 usersLen = _users.length;
          pendings = new uint256[][](usersLen);
          uint256 poolLen = poolInfo.length;
          for(uint256 i = 0; i < usersLen; i++) {
              address _user = _users[i];
              pendings[i] = new uint256[](poolLen);
              for(uint256 pid = 0; pid < poolLen; ++pid) {
                  pendings[i][pid] = _pendingPoints(pid, _user);
              }
          }
      }
  }